Send a mail when a cell reaches a certain value 

To run a macro automatic when a specific cell reaches a certain value you can use
the Change event in the worksheet module.
The examples on this page use Cell A1 and will run the macro if the cell value > 200
 

 

Change Event
 

1) Right click on a sheet tab and choose view code
2) Paste one of the events in this module.
3) Alt-Q to go back to Excel
 

 

If you change the cell manual you can use this example for A1.

Change YourMacroName to the name of your macro.

 

	Private Sub Worksheet_Change(ByVal Target As Range)

    If Not Application.Intersect(Range("A1"), Target) Is Nothing Then

        If IsNumeric(Target.Value) And Target.Value > 200 Then

            YourMacroName

        End If

    End If

End Sub


 

 

If A1 is a formula you can use this example
 

	Private Sub Worksheet_Change(ByVal Target As Range)

    Dim rng As Range

    If Target.Cells.Count > 1 Then Exit Sub

    On Error GoTo EndMacro

    If Not Target.HasFormula Then

        Set rng = Target.Dependents

        If Not Intersect(Range("A1"), rng) Is Nothing Then

            If Range("A1").Value > 200 Then YourMacroName

        End If

    End If

EndMacro:

End Sub


 

 

 

Example macro's
 

Note : copy the macro's in a normal module
 

 

Example 1
 

This is the best way if you can use it?
Read the information on this page first.
http://www.rondebruin.nl/cdo.htm
 

 

	Sub Mail_CDO()

    Dim iMsg As Object

    Dim iConf As Object

    '    Dim Flds As Variant
    Set iMsg = CreateObject("CDO.Message")

    Set iConf = CreateObject("CDO.Configuration")
    '    iConf.Load -1    ' CDO Source Defaults

    '    Set Flds = iConf.Fields

    '    With Flds

    '        .Item("http://schemas.microsoft.com/cdo/configuration/sendusing") = 2

    '        .Item("http://schemas.microsoft.com/cdo/configuration/smtpserver") = "Fill in your SMTP server here"

    '        .Item("http://schemas.microsoft.com/cdo/configuration/smtpserverport") = 25

    '        .Update

    '    End With
    With iMsg

        Set .Configuration = iConf

        .To = "ron@debruin.nl"

        .CC = ""

        .BCC = ""

        .From = """Ron"" <ron@something.nl>"

        .Subject = "Important message"

        .TextBody = "Hi there" & vbNewLine & vbNewLine & _

                    "Cell A1 is changed"

        .Send

    End With

    Set iMsg = Nothing

    Set iConf = Nothing

End Sub


 

 

 

Example 2
 

You can use this example only if you use Outlook.
 

If you don't like the security warning in Outlook see this page
http://www.rondebruin.nl/mail/prevent.htm
 

 

	Sub Mail_with_outlook()

    Dim OutApp As Object

    Dim OutMail As Object

    Dim strto As String, strcc As String, strbcc As String

    Dim strsub As String, strbody As String
 

    Set OutApp = CreateObject("Outlook.Application")
    OutApp.Session.Logon

    Set OutMail = OutApp.CreateItem(0)
 

    strto = "ron@debruin.nl"

    strcc = ""

    strbcc = ""

    strsub = "Important message"

    strbody = "Hi there" & vbNewLine & vbNewLine & _

              "Cell A1 is changed"
 

    With OutMail

        .To = strto

        .CC = strcc

        .BCC = strbcc

        .Subject = strsub

        .Body = strbody

        .Send

    End With
 

    Set OutMail = Nothing

    Set OutApp = Nothing

End Sub


 

 

 

Example 3
 

This example is for Outlook Express.
 

If you don't like the security warning in Outlook Express see this page
http://www.rondebruin.nl/mail/prevent.htm
 

 

	Sub Mail_Outlook_Express()

    Dim Recipient As String, Subj As String, HLink As String

    Dim Recipientcc As String, Recipientbcc As String

    Dim msg As String
    Recipient = "ron@debruin.nl"

    Recipientcc = ""

    Recipientbcc = ""

    Subj = "Important message"

    msg = "Hi there" & vbNewLine & vbNewLine & _

          "Cell A1 is changed"
    msg = WorksheetFunction.Substitute(msg, vbNewLine, "%0D%0A")

    HLink = "mailto:" & Recipient & "?" & "cc=" & Recipientcc _

          & "&" & "bcc=" & Recipientbcc & "&"

    HLink = HLink & "subject=" & Subj & "&"

    HLink = HLink & "body=" & msg
    ActiveWorkbook.FollowHyperlink (HLink)

    Application.Wait (Now + TimeValue("0:00:02"))

    Application.SendKeys "%s"

End Sub


 

